
Problema 2, Primera Ronda CyM 2000, Nivel 1:Problema 2, Primera Ronda CyM 2000, Nivel 1:Problema 2, Primera Ronda CyM 2000, Nivel 1:Problema 2, Primera Ronda CyM 2000, Nivel 1:    

Una acaudalada anciana tiene gran cantidad de lingotes de oro de tres tipos: por 
valor de $56, de $106, y de $127. ¿Cuál es la menor cantidad de lingotes que puede 
usar para depositar exactamente $5409 en una caja de seguridad de un banco?  

 

En papel 
Primero vamos a escribir la ecuación en 

papel. Llamamos x a la cantidad de lingotes de 
$56 que deposita la señora. Así que en lingotes 
chiquitos deposita x · 56 pesos. 

De la misma manera  llamamos y a la 
cantidad de lingotes de $106 y z a la cantidad 
de lingotes de $127. 

En total la anciana deposita en el banco 
x · 56 + y · 106 + z · 127 pesos. El total que 
deposita es de 5409 así que queda para 
resolver   x · 56 + y · 106 + z · 127 = 5409 . 

Notamos que x no puede ser muy grande, 
porque si usara por ejemplo 1000 lingotes de 
$56, el depósito sería demasiado grande. La 
máxima cantidad de lingotes pequeños que 
puede utilizar es 5409/56 = 96,589..., o sea que x 
está entre 0 y 96 (ambos extremos incluidos). 

De la misma manera y está entre 0 y 51 
porque 5409/106 = 51,028... y z  está entre 0 y 42 
porque 5409/127 = 42,590... . Así que hay que 
probar sólo 97 · 52 · 43 ≅ 200000 posibilidades, 
que para la computadora son pocas. 

Vamos a escribir un programa que 
encuentra todas las combinaciones de x, y y z 
que cumplen la ecuación y después vamos a 
elegir a mano la que utiliza menos lingotes. 

El programa en QB 
Dim X as Long 

Dim Y as Long 

Dim Z as Long 

Print " X", " Y", " Z", " Total" 

For X = 0 To 97 

  For Y = 0 To 51 

    For Z = 0 To 42 

      If X*56+Y*106+Z*127=5409 Then 

        Print X, Y, Z, X + Y + Z 

      End If 

    Next Z 

  Next Y 

Next X 
 

En la pantalla 
(No hace falta copiarlo en papel.) 

 X     Y     Z     Total 

 2     32    15    49 

 5     40    7     52 

 7     3     37    47 

 10    11    29    50 

 13    19    21    53 

 16    27    13    56 

 19    35    5     59 

 24    6     27    57 

 27    14    19    60 

 30    22    11    63 

 33    30    3     66 

 38    1     25    64 

 41    9     17    67 

 44    17    9     70 

 47    25    1     73 

 55    4     15    74 

 58    12    7     77 

 72    7     5     84 

 86    2     3     91 
 

En papel nuevamente 
(Es importante escribirlo en papel.) 

El programa muestra en cada caso el total 
de lingotes para poder buscar más fácilmente 
el menor a ojo. 

Mirando la pantalla buscamos la 
combinación con la menor cantidad de 
lingotes que puede usar la señora : 7 lingotes 
de $56, 3 lingotes de $106 y 37 lingotes de $127.  

Comentarios al margen 
También se puede hacer que la 

computadora elija automáticamente la 
combinación con menos lingotes.  

Otra posibilidad es usar dos ciclos for para 
elegir x e y. En cada caso despejar z y fijarse si 
es un número entero. Esto anda un poco más 
rápido por que no prueba con todos los valores 
de z, uno por uno. 

Recuerden escribir la respuesta del problema 
en papel (con letra prolija).  

Problema 1, Primera Ronda CyM 1999, Nivel 1:Problema 1, Primera Ronda CyM 1999, Nivel 1:Problema 1, Primera Ronda CyM 1999, Nivel 1:Problema 1, Primera Ronda CyM 1999, Nivel 1:    

¿Cuántos números enteros positivos menores que 1020 tienen como únicos 
factores primos al 2, 3 ó 7? (Por ejemplo: 2, 8, 21, 63, 84, ...) 

(Nota: los números enteros positivos son 1, 2, 3, 4, ...) 
 

En borrador 
Primero tratamos de entender a fondo el 

problema. Llamemos n a un número que 
cumpla con lo que pide el enunciado. 
Entonces, como es “entero positivo”, debe ser 
n ≥ 1. También el enunciado dice que n < 1020. 

La condición que falta es que n tiene como 
únicos factores primos al 2, al 3 y/o al 7. Pruebo 
con los ejemplos del enunciado: 

8 = 2·2·2 = 23 
21 = 3·7 
63 = 3·3·7 = 32·7 
Y también con otros números: 
17 = ... = 17 (es primo, no sirve) 
35 = 5·7 (tiene el 5, no sirve) 
6 = 2·3 (¡este sí!) 
O sea, si escribo n como producto de sus 

factores primos, debe ser n = 2a·3b·7c. 
(A continuación describimos una solución 

posible. No es la más eficiente, pero funciona, y 
quizás sea lo primero que se nos ocurre.) 

En papel 
El enunciado pide contar cuántos enteros n 

cumplen n ≥ 1, n < 1020, y n = 2a·3b·7c.  
Como el 1 no sirve, lo descartamos. 
Lo que vamos a hacer es un programa que 

prueba todos los números enteros desde 2 
hasta 1019, y los cuente cuando el número sólo 
tiene como factores primos al 2, 3 ó 7. 

Primero hacemos una copia del valor del 
Numero en la variable Queda. 

Vamos a eliminar todos los 2 de la 
factorización de Queda. Si tiene de factor al 2 
al dividirlo por 2 da resto 0, calculamos el resto 
con la función “mod”. Si es así la dividimos por 
2. Repitiendo este proceso muchas veces 
estamos seguros de eliminar a todos los 2. 
Como los números son menores a 1020 alcanza 
con repetir este procedimiento 1020 veces. 

(Se puede achicar mucho este número, pero 
no hace falta, porque el programa es rápido. 
Para cada número el programa tiene que 
hacer unas 1020 cuentas tres veces (para 2, 3 y 
7). Como hay 1018 numero tiene que hacer sólo 
unas 1020*3*1018 ≅ 3000000 operaciones.) 

Luego hacemos lo mismo con 3 y con 7. 
Si al final Queda vale 1, significa que no 

aparecen otros primos en la factorización de n 
y entonces es de la forma 2a·3b·7c y lo 
contamos. Si no, quiere decir que todavía 
quedan otros factores primos, y no lo contamos. 

El programa en QB 
Dim Cantidad as Long 

Dim Numero as Long 

Dim Queda as Long 

Dim Aux as Long 

Cantidad = 0 

For Numero = 2 To 1019 

    Queda = Numero 

    For Aux = 1 To 1020 

        If Queda mod 2 = 0 Then 

            Queda = Queda / 2 

        End If 

    Next Aux 

    For Aux = 1 To 1020 

        If Queda mod 3 = 0 Then 

            Queda = Queda / 3 

        End If 

    Next Aux 

    For Aux = 1 To 1020 

        If Queda mod 7 = 0 Then 

            Queda = Queda / 7 

        End If 

    Next Aux 

    If Queda = 1 Then 

        Cantidad = Cantidad + 1 

    End If 

Next Numero 

Print "El resultado es:", Cantidad  

En la pantalla 
El resultado es: 74 

En papel nuevamente 
(Es importante escribirlo en papel.) 

El programa muestra directamente el 
resultado: hay 74 enteros positivos menores que 
1020 cuyos únicos factores primos son 2, 3 ó 7. 

Comentarios al margen 
A veces es conveniente ir mostrando los 

números que va encontrando, para estar 
seguros de haber hecho bien las cosas. 

Hay varias formas distintas de resolver y de 
escribir un programa que realice esta tarea. Por 
ejemplo se puede hacer que cuando se da 
cuenta que no es múltiplo de 2, no siga 
probando inútilmente. 

Otra posibilidad, en lugar de probar número 
por número si sirve o no, es simplemente 
generarlos (con cuidado). Para esto usamos 
que n = 2a · 3b · 7c, y mediante ciclos anidados 
vamos dando valores a a, b y c, verificando 
que el producto (n) sea menor que 1020. Notar 
que a, b y c no pueden ser muy grandes. Con 
este método anda más rápido. 

Recuerden escribir la respuesta del problema 
en papel (con letra prolija). 

